(7-57-M)

1. IDENTIFICATION

1.1 Maindec 710

1.2 PDP-7 Tape Reader (Type 444B) Test

1.3 August 27, 1965

PDP 7 LIBRARY

NOTE: Change code for variable word size listing

2. ABSTRACT

Maindec 710 incorporates two separate test programs. The first tests the performance of all the IOT instructions associated with the reader and its connections to the program interrupt. It reads a loop of tape containing alternate characters of 1's and 0's, first in alphanumeric, then in binary mode.

The second test reads a loop of tape containing the sequence 1-377 repeated three times. This program tests only the accuracy of reading; the interrupt is not used, and the flags are assumed to be working.

3. REQUIREMENTS

3.1 Storage

Maindec 710 occupies memory locations 0 - 7 and 20 - 1325. It requires that the RIM Loader be in memory.

3.2 Subprograms and/or Subroutines

Maindec 710 uses two library subroutines:

Teletype Output Package (Digital-7-10-0)
Octal Print Subroutine (Digital-7-14-0)

Both subroutines are included in the Maindec 710 Funny Format binary tape provided. The ASCII tapes of the two subroutines are not supplied.

3.3 Equipment

Standard PDP-7 with paper tape reader and punch.

3.4 Miscellaneous

Tapes supplied:

ASCII (Test program only)
Funny Format binary

put in a rim loader

4. USAGE

4.1 Loading

4.1.1 Place the FF binary tape in the reader.

4.1.2 Set the ADDRESS switches to (1)7770.

4.1.3 Press START.

4.3 Switch Settings

4.3.1 Loading Address: (1)7770

4.3.2 Starting Addresses

Section 1:	100	(To punch test loop)
	140	(To test reader)
Section 2:	200	(To punch test loop)
	375	(To test reader)

146 eliminates type outs.

4.3.3 Other Switch Settings

The settings given apply to both sections of the test program.

Switch	Setting	Function
ACS_0	Down	Halt on error
	Up	No halt on error
ACS_1	Down	Print error messages
	Up	Do not print message
ACS_{2-17}	---	The setting determines the speed at which the reader operates: the higher the number in the switches, the slower the rate used down to a minimum of 3 cps.

4.4 Start up and/or Entry

4.4.1 Section 1: Maximum Noise Test

Set the ADDRESS switches to 00100 and press START. The program title will be printed on the teleprinter:

```
SECTION 1 TEST.
POSITION TAPE IN READER.
ACS0 DOWN FOR HALTS, UP FOR NONE.
ACS1 DOWN FOR PRINTING, UP FOR NONE.
ACS2-17 FOR SPEED CONTROL.
      CHARACTER
READ          EXPECTED
```

The computer will stop with the AC and link cleared.

Remove the test tape from the punch bin, and make a loop by overlapping the blank ends of the tape.

Place the test loop in the reader so that the blank section falls under the read heads. Put the reader arm down.

Set the AC switches as desired (see Section 4.3.3) and press CONTINUE. The program will read the loop and continue until an error is encountered or until the operator stops the computer.

It is not necessary to punch a new loop every time the test is run. If a loop has already been made, place it in the reader, set the ADDRESS switches to 140, and press START. The title and instructions will be printed, and the computer will halt. Proceed as outlined above.

4.4.2 Section 2: Numeric Sequence Test

Set the ADDRESS switches to 00200 and press START. The title will be printed:

```
SECTION 2 PUNCH.
```

The test tape is punched, and the title and instructions for the test are printed. These are identical to the instructions for Section 1, except for the first line, which reads:

SECTION 2 TEST.

The computer halts with the AC and link clear. Make a loop of the test tape by overlapping blank ends as before, and place it in the reader with the blank section under the read heads. Put the reader arm down. Set the AC switches as desired, and press CONTINUE.

The test will run until an error is encountered, or until the operator stops the computer.

As in Section 1, the test tape can be used over again. If the loop has been prepared, place it in the reader as before. Set the ADDRESS switches to 00375, and press START. The title and instructions are printed, and the procedure is as described above.

4.5 Errors in Usage

On all error halts, the link is set to 1.

Error E01-E05 will halt only if ACS_0 is down.

Error:	E01
Message:	CAF DID NOT CLEAR FLAG.
C(MA):	00153
C(AC):	I/O status
Cause:	The CAF instruction failed to clear the reader flag. Since the flag is not set by the program up to this point, the error implies that the flag is permanently on; otherwise, it would have been cleared when the START button was pressed to start the program.
Recovery:	Press CONTINUE. The test will proceed from the point of error.
Error:	E02
Message:	RSF SKIPPED ON CLEARED FLAG.
C(MA):	00047
C(AC):	I/O status
Cause:	The reader flag was clear; nevertheless the RSF instruction skipped.
Recovery:	Press CONTINUE. The test will proceed from the point of error.
Error:	E03
Message:	RSA DID NOT CLEAR FLAG. or RSB DID NOT CLEAR FLAG.
C(MA):	00226
C(AC):	I/O status
Cause:	In both cases, the reader select instruction failed to clear the reader flag. This may imply the same thing as E01, above, unless it is clear that the tape has been read at least in part, in which case this is not likely to be a systematic failure.
Error:	E04
Message:	RRB DID NOT CLEAR FLAG.
C(MA):	00246
C(AC):	I/O status
Cause:	Again, the flag was unaffected by an associated IOT.
Recovery:	Press CONTINUE. The test will proceed from the point of error.

Error: E05
Message: RSF DID NOT SKIP WHEN FLAG WAS SET.
C(MA): 00032
C(AC): I/O status
Cause: This is the other possible skip error. In this case, the RSF failed to skip even though the flag was set.
Recovery: Press CONTINUE. The test will proceed from the point of error.

Errors E30-E33 will always halt, regardless of the setting of ACS_0.

Error: E30
Message: RRB DID NOT CLEAR AC.
C(MA): 00337
C(AC): 777777
Cause: The AC was not cleared when the RRB instruction was executed.
Recovery: Reposition the tape so that the blank section falls under the read heads, and put the arm down. Press CONTINUE. The program will try to put the tape in phase again.

Error: E31
Message: LEADER SIGNAL IS INCORRECT.
C(MA): 00332
C(AC): Word read from tape.
Cause: On both test tapes, the test sequence is preceded by a signal to indicate the end of the blank tape section. This signal consists of three binary characters forming the word 770000. The program tests to see if the signal read from tape is in fact 770000. If not, this error occurs.
Recovery: Reposition the tape so that the blank section falls under the read heads, and put the arm down. Press CONTINUE. The program will try to put the tape in phase again.

Error: E32
Message: NO INTERRUPT AFTER 10 MSEC. IS THE READER ARM DOWN?
C(MA): 00236
C(AC): I/O status
Cause: The time ran out while waiting for a program interrupt from the reader. The question in the diagnostic implies that the tape did not move beyond the leader signal. If the error occurs after some of the test sequence has been read, it may be the result of a failure in the program interrupt control.
Recovery: Press CONTINUE. The test will begin again, printing the title and instructions. When the computer halts, proceed as described in Section 4.4.1 or 4.4.2.

In each of the following errors, no message is printed. Instead, the erroneous and correct characters are printed in the columns labeled "READ" and "EXPECTED," respectively. Errors E10 and E11 refer to the alphanumeric sequence of the Section 1 test. Errors E12 and E13 refer to the binary sequence of the same test. Error E20 refers to Section 2. Each of these errors occurs when the character (or word) read from tape does not match that expected by the program. In each case, when the program halts, the AC displays as 1's those bits which do not match.

Error: E10
C(MA): 00166
Expected Character: 377

Error:	E11
C(MA):	00174
Expected Character:	000
Error:	E12
C(MA):	00211
Expected Word:	770077
Error:	E13
C(MA):	00217
Expected Word:	007700
Error:	E20
C(MA):	00443
Expected Character:	1-377 in sequence.
Recovery:	From each of these error halts, the recovery procedure is the same. Press CONTINUE. The test will proceed from the point of error.
Error:	E33
Message:	INTERRUPT FROM SOMETHING OTHER THAN THE READER. CHECK THE OTHER DEVICES...I/O STATUS IS IN THE AC.
C(MA):	00026
C(AC):	I/O status
Cause:	A device other than the reader has caused an interrupt. By examining the I/O status, the offending device probably can be identified.
Recovery:	Press CONTINUE. The test will begin again, printing the title and instructions. When the computer halts, proceed as described in Section 4.4.1 or 4.4.2.

4.6 Recovery from Such Errors:

See Section 4.5, Errors.

5. RESTRICTIONS

If the automatic priority interrupt option is installed, it must be kept off. Since pressing the START key disables the API, there should be no interference from that system.

6. DESCRIPTION

6.1 Section 1

The first section of Maindec 710 tests all of the IOTs associated with the tape reader, the performance of the program interrupt, and the accuracy of information transfer under conditions of maximum noise. This is obtained by causing the reader to scan a tape punched with alternating characters of 337 and 000.

(Reading in binary mode, the alternation is between words of 770077 and 007700.) The test tape is produced by an auxiliary punch program. The tape has the following format:

blank leader (40 lines)
leader signal
alphanumeric sequence
binary sequence
blank trailer (40 lines)

The leader signal consists of the three binary characters 277, 200, 200. The alphanumeric sequence consists of 257_{10} character pairs of 377 and 000. The binary sequence consists of 258_{10} character pairs of 277 and 200. The test program keeps a count as it scans the tape. When the end is reached, an RSB is given which causes the blank section of the loop to be skipped. The next information read from tape is the leader signal, and the test begins again.

The reading speed is controlled by the setting of AC switches 2-17. If all switches are down, there is no delay, and the reader operates at maximum speed, 300 cps. The speed decreases linearly as the value of the switches increases. The slowest speed is obtained by a setting of 177777, corresponding to a reading rate of about 3 cps.

6.1.2 Section 2

This test operates in exactly the same manner as Section 1, except that it assumes that all IOTs are working (except, possibly, RSA). It does not use the program interrupt. The tape has the same leader and trailer format and leader signal. The body of the tape consists of the sequence of character codes 1-377 repeated three times. This test is designed to check the accuracy of transmission of every possible combination of bits.

The speed control works exactly the same way as for Section 1.

6.2 Examples and/or Applications

The use of each section of Maindec 710 is evident from the foregoing. It should be noted, however, that the speed control can be used to detect certain types of failures. There are certain reading rates at which the reader is more likely to fail than at others. Using the ACS, the operator can find these error points by making the following test.

Start the program with the loop positioned and ACS_{2-17} down. After a short time, raise ACS_{17}. Gradually decrease the reading speed by putting up each switch in turn from the lowest order to the highest, putting the previous switch down as you raise the next. At some point, the reader may fail. You can then test reader accuracy at speeds near that point.

7. METHODS

See DESCRIPTION

8. FORMAT (Not Applicable)

9. EXECUTION TIME

The time for one pass over a test tape depends on the setting of the AC switches. Reading rates can be varied from 3 to 300 cps.

10.4 Program Listing

```
/MAINDEC 710: PDP-7 TAPE READER TEST
/IN TWO SECTIONS: SECTION 1---MAXIMUM NOISE, RSA, ^RSB
                  /SECTION 2--NUMERIC SEQUENCE, RSA ONLY
100/                                    /SECTION 1 PUNCH
PUNT1,      TIN
            LAW PIT                     /TITLE
            TSR
            LAM -50
            JMS FEED                    /LEADER
            JMS PULS                    /LEADER SIGNAL
PALF,       LAM -400                    /ALPHA SEQUENCE
            DAC APTEM
            LAW 377
            PSA
            PSF
            JMP .-1
            SMAV CMA                    /COUNT?
            ISZ APTEM                   /YES
            JMP .-5
PUBS,       LAM-401                     /BINARY SEQUENCE
            DAC APTEM
            LAW 77
            JMS BIP
            SMA  CMA                    /COUNT?
            ISZ APTEM                   /YES
            JMP .-3
PEND,       LAM-50
            JMS FEED
            JMP RET1                    /GO TEST
PULS,       0                           /PUNCH LEADER SIGNAL
            LAW 77
            JMS BIP
            CLA
            JMS BIP
            JMS BIP
            JMP I PULS
ORB-700102                              /OR READER BUFFER INTO AC.
/MAINDEC 710, PAGE 2
RET1,       TIN                         /SECTION 1 TEST
            LAW R1T
            TSR
            LAW RIP
            TSR
A,          HLTVCLAVCLL
            LAC DCAF                    /INITIALIZE EM0
```

```
                DAC DIPT
                CLOF
                CAF
                JMS REFT        /TEST FLAGS
E01,            HLTV STL        /CAF DIDN'T CLEAR
INRA,           JMS TAPH        /INITIALIZE TEST
                LAC DRSA
                DAC DIPT
                LAM -400
                DAC APTEM
RALPH,          JMS DELAY       /ALPHA TEST
                RSA
163             LAM -4000       CLA   /INTERRUPT TIME
                JMS TESC
                377             /ARGUMENT
166 E10,        HLTV STL        /BAD BITS IN AC ON HALT
                JMS DELAY
                RSA
171             LAM -4000       CLA
                JMS TESC
                000
E11,            HLT  STL
                ISZ APTEM
                JMP RALPH
                SKP             /GET AROUND ENTRY
200/            JMP PUNT2       /ENTRY FOR SECTION 2
INREB,          ISZ DIPT        /CHANGES TO RSB
                LAM -125
                DAC APTEM
204 BIRD,       JMS DELAY       /BINARY TEST
                RSB
206             LAM -14000      CLA   /BINARY MODE TIMER
                JMS TESC
                770077
211 E12,        HLTV STL
                JMS DELAY
                RSB
214             LAM -14000      CLA
                JMS TESC
                007700
E13,            HLTV STL
/MAINDEC 710:  PAGE 3
                ISZ APTEM       /FINISHED?
                JMP BIRD        /NO
                JMP INRA        /YES.  ROUND AGAIN
TESC,           0               /COMPARING SUBROUTINE
                DAC PICT        /SET TIMER
```

```
            JMS REFT
E03,        HLTVSTL             /FLAG STILL SET
            ION
            ISZ PICT            /TIME TILL BREAK
            JMP .-1

ER32,       IORS                /NO BREAK
            JMS RACS
            LAW EM32
            NOP                 /ALWAYS HALTS
E32,        HLTVSTL
            JMP RET1

IRET,       IORS                /RETURN HERE FROM BREAK
            RAL
            SMA  RAR            /IS THE FLAG CLEAR?
            JMP COMC            /YES.
            JMS RACS            /NO.
            LAW EM04
E04,        HLTVSTL

COMC,       LAC AP2
            SAD I TESC          /COMPARE CHARACTERS
            JMP TOUT            /OK.
ER1,        JMS RACS            /BAD.
            JMP ERP
            JMP E1H
TOUT,       ISZ TESC            /GO AWAY
            ISZ TESC
            JMP I TESC

E1H,        LAC I TESC          /ERROR STOP
            XOR AP2             /EXTRACT BAD BITS
            JMP TOUT+1

ERP,        TIN                 /ERROR PRINT
            LAC AP2
            JMS OPT
            TYT
            LAC I TESC
            JMS OPT
            JMP RAC2

STEM,       0                   /STATUS HOLDER
PICT,       0                   /INTERRUPT TIMER

RACS,       0                   /SAVE THIS SPACE

/MAINDEC 710, PAGE 4

/INTERRUPT SERVICES AND OTHERS

0/          0
            IORS
            RAL
```

```
                SMAV RAR            /IS READER FLAG SET?
                JMP ER33            /NO
                RSF                 /TEST SKIP
                JMP ER05            /FAILED
                JMP IAWAY

20/             0                   /CALCATCHER
                HLT
                JMP .-1

ER33,           JMS RACS            /SPURIOUS INTERRUPT
                LAW EM33
                NOP                 /ALWAYS HALTS
E33,            HLTV STL
                JMP RET1

ER05,           JMS RACS            /RSF FAILED
                LAW EM05
E05,            HLTV STL

IAWAY,          RRB                 /OK, READ BUFFER
                DAC AP2
                JMP IRET            /GO COMPARE

REFT,           0                   /FLAG TEST SUBROUTINE
                IORS
                RAL
                SPAV RAR            /IS FLAG STILL SET?
                JMP ER0             /YES
                RSF                 /NO.  TEST SKIP
                JMP REO             /OK.  GO AWAY
ER02,           JMS RACS            /SKIPPED ON CLEAR FLAG
                LAW EM02
E02,            HLTV STL
                JMP REO

ER0,            JMS RACS            /FLAG NOT CLEARED
                JMP E0P
                JMP I REFT          /IF STOPPING
REO,            ISZ REFT            /IF NOT
                JMP I REFT

E0P,            LAC DIPT            /PRINT DIAGNOSTIC
                TY3
                LAW EM0
                JMP RAC2-1

/MAINDEC 710, PAGE 5

/VARIOUS KINDS OF SUBROUTINES

RACS/           0                   /DIAGNOSTIC HANDLER
                DAC STEM            /SAVE STATUS
                IOF
                LAS
```

```
                RAL
                SPAV RAR            /PRINT MESSAGE?
                JMP RAC2+1          /NO
                TIN
                XCT I RACS          /LAW OR JMP
                TSR                 /IF LAW
RAC2,           LAS
                SPA                 /STOP FOR ERROR?
                ISZ RACS            /NO
                ISZ RACS
                CAF                 /CLEARS HANGING TTY FLAGS
                LAC STEM
                JMP I RACS

TAPH,           0                   /TAPE PHASER
                RSB                 /SKIP BLANK TAPE, PICK UP LEAD SIGNAL
                RSF
                JMP .-1
                CLC                 /TO TEST RRB
                RRB
                SAD (-0             /DID AC CLEAR?
                JMP ER30            /NO
                SAD (770000         /IS SIGNAL OK?
                JMP I TAPH          /YES, PROCEED

ER31,           JMS RACS            /NO
                LAW EM31
                NOP                 /ALWAYS HALTS
E31,            HLTV STL
                JMP TAPH+1          /TRY AGAIN

ER30,           JMS RACS            /RRB FAILED
                LAW EM30
                NOP                 /THIS ONE ALWAYS STOPS, TOO
E30,            HLTV STL
                JMP TAPH+1          /TRY AGAIN

FEED,           0                   /TAPE BLANKER
                DAC APTEM
                PSA+10
                PSF
                JMP .-1
                ISZ APTEM
                JMP .-4
                JMP I FEED

/MAINDED 710:  PAGE 6

/SECTION 2---NUMERIC SEQUENCE TEST

PUNT2,          TIN                 /SECT. 2 PUNCH
                LAW P2T
                TSR
                LAM -50
```

```
                JMS FEED
                JMS PULS
PUSE,           LAM -2
                DAC APTEM       /THREE CYCLES OF 1-377
                LAM -376
                DAC PIX
PIX,            LAM             /INDEX AND LOAD AC
                PSA
                PSF
                JMP .-1
                ISZ PIX         /END OF CYCLE
                JMP PIX         /NO
                ISZ APTEM       /YES - END OF SEQUENCE?
                JMP PIX-2       /NO
                LAM -50         /YES
                JMS FEED
RET2,           TIN             /SECTION 2 TEST
                LAW R2T
                TSR
                LAW RIP
                TSR
                IOF
B,              HLTVCLAVCLL
                JMS TAPH        /TRY PHASING TAPE
                LAM -2          /SET UP COUNTERS
                DAC AP2
NESQ,           LAM -376
                DAC APTEM
                JMS DELAY
RECH,           RSA
                RSF
                JMP .-1
                LAM -377        /777400 IN AC
                ORB             /ORS READER BUFFER INTO AC
                SAD APTEM       /COMPARE RESULT
                JMP REX         /OK
ER20,           JMS RACS        /NO GOOD
                JMP E20P
                JMP E20H
/MAINDEC 710:  PAGE 7
REX,            ISZ APTEM       /END OF CYCLE?
                JMP RECH-1      /NO
                ISZ AP2         /YES.  END OF TAPE?
                JMP NESQ        /NO
                JMP B+1         /YES.  GO ROUND AGAIN
E20P,           TIN
                LAC STEM
```

```
                AND (377                /EXTRACT CHARACTER
                JMS OPT
                TYT
                LAC APTEM
                AND (377
                JMS OPT
                JMP RAC2
E20H,           XOR APTEM
E20,            HLTV STL                /AC SHOWS WRONG BITS
                JMP REX
DELAY,          0                       /SPEED CONTROL
                LAS
                AND (177777
                CMA
                DAC AP3
                ISZ AP3
                JMP .-1
DLOUT,          JMP I DELAY
E0P+4/
OTY,            0                       /PRINT A CHARACTER
                TLS
                TSF
                JMP .-1
                JMP I OTY
APTEM,          0                       /ALL PURPOSE TEMPORARY STORAGE
AP2,            0
AP3,            0
BIP,            0                       /PUNCH BINARY CHARACTER
                PSB
                PSF
                JMP .-1
                JMP I BIP
/MAINDEC 710:  PAGE 8
/ORACLES, WARNINGS, AND ANSWERS
DLOUT+1/
P1T,            TEXT -SECTION 1 PUNCH.
-
R1T,            TEXT -SECTION 1 TEST.
-
P2T,            TEXT -SECTION 2 PUNCH.
-
R2T,            TEXT -SECTION 2 TEST.
-
RIP,            TEXT                    /POSITION TAPE IN READER.
ACS0 DOWN FOR HALTS, UP FOR NONE.
```

```
ACS1 DOWN FOR PRINTING, UP FOR NONE
ACS2-17 FOR SPEED CONTROL.

     CHARACTER
READ         EXPECTED

DCAF,        FLEX CAF              /DIPT SETUPS
DRSA,        FLEX RSA
DIPT,        0
EM0,         TEXT -DID NOT CLEAR FLAG.
-
EM02,        TEXT -RSF SKIPPED ON CLEARED FLAG.
-
EM04,        TEXT -RRB DID NOT CLEAR FLAG.
-
EM05,        TEXT -RSF DID NOT SKIP WHEN FLAG WAS SET.
-
EM30,        TEXT -RRB DID NOT CLEAR AC.
-
EM31,        TEXT -LEADER SIGNAL IS INCORRECT.
-
EM32,        TEXT -NO INTERRUPT AFTER 10 MSEC.
IS THE READER ARM DOWN?
-
EM33,        TEXT -INTERRUPT FROM SOMETHING OTHER THAN THE READER.
CHECK THE OTHER DEVICES...I/O STATUS IS IN THE AC.
-

START

/TELETYPE OUTPUT PACKAGE 8-13-63

XIT=LAC-JMS   TTAB=10

/TYPE 1 CHARACTER FROM AC BITS 12-17

TY1=JMS .
              0
              DAC TY+ SVAC
              RAR
              JMS TY1A
              XIT TY1
              TYEXIT

/TYPE 1 CHARACTER (5 BIT), LINK INDICATES CASE

TY1A,         0
              DAC T+ EMY
              AND (37
              SNA
              JMP TY2
              703301
              SKP
              JMP TY1B
```

```
                LAC OCL
                SZL
                LAC OCU
                SAD OCS
                JMP . 3
                JMS OTY
                DAC OCS
                LAC TEMY
                JMS OTY
                ISZ T  BC
TY2,            LAC TEMY
                JMP I TY1A
```

/TYPE 3 CHARACTERS FROM AC 0-5, 6-11, 12-17 RESPECTIVELY

```
TY3=JMS .
                0
                DAC TYSVAC
                JMS RL6
                JMS TY1A
                JMS RL6
                JMS TY1A
                JMS RL6
                JMS TY1A
                XIT TY3
                TYEXIT
```

/TELETYPE OUTPACKAGE PAGE 2

/TYPE A CARRIAGE RETURN, AND LINE FEED

```
TCR=JMS .
                0
                DAC TYSVAC
                703301
                SKP
                JMP TCRA
                LAW 2
                JMS OTY
                LAW 10
                JMS OTY
                DZM TBC
                XIT TCR
                TYEXIT
```

/TYPE A SPACE

```
TSP=JMS .
                0
                DAC TYSVAC
                LAW 4
                703301
                SKP
```

```
                LAW 240
                JMS OTY
                ISZ TBC
                XIT TSP
                TYEXIT

/TYPE A TABULATION

TYT=JMS .
                0
                DAC TYSVAC
                LAC TBC
                ADD (1
                TAD (-TTAB
                SMA
                JMP .-2
                TAD (-1
                DAC TEMY1
                LAC TYSVAC
                TSP
                ISZ TEMY1
                JMP .-2
                XIT TYT
                TYEXIT

/TELETYPE OUTPUT PACKAGE - PAGE 3

/TYPEWRITER INITIALIZE

TIN=JMS .
                0
                DAC TYSVAC
                LAC OCL
                DAC OCS
                703301
                SKP
                JMP . 3
                TLS
                JMS OTY
                LAC TYSVAC
                TCR
                JMP I TIN-JMS

/TYPE THE DIGIT IN THE AC

TDIGIT=JMS .
                0
                DAC TEMY1
                AND (17
                ADD (LAC NCT
                DAC . 1
                XX
                TY1
```

```
                LAC TEMY1
                JMP I TDIGIT-JMS
/TYPE A STRING OF CHARACTERS
TSR=JMS .
                0
                DAC T+EMY1
                LAC I TEMY1
                TY3
                AND (76
                ISZ TEMY1
                SZA
                JMP TSR+2-JMS
                LAC TEMY1
                JMP I TSR-JMS
/EXIT AFTER RESTORING AC AND LINK
                TYEXIT=JMP .
                DAC TEMY
                RAL
                LAC TYSVAC
                JMP I TEMY
/TELETYPE OUTPUT PACKAGE - PAGE 4
/ROTATE LEFT 6
RL6,            0
                RTL
                RTL
                RTL
                JMP I RL6
/TABLE OF DIGITS
NCT,            33      73      63      41      25
                3       53      71      31      7
/CASE STORAGE
OCU,            33
OCL,            37
OCS,            0
/END OF TELETYPE OUTPUT PACKAGE
/PDP-4/7 ADDENDUM
TY1B,           ADD (LAC BTATAB-1
                DAC . 1
                XX
                SZL
                JMP TY1C
TY1D,           JMS OTY
                JMP TY2-1
```

```
TY1C,           JMS RL6
                RTL
                RTL
                JMP TY1D
TCRA,           LAW 215
                JMS OTY
                LAW 212
                JMP TCR-JMS 10
BTATAB,         265324          /5,T
                215215          /CARRIAGE RETURN
                271317          /9,O
                240240          /SPACE
                243310          /x,H
                254316          /,,N
                256315          /.,M
                212212          /LINE FEED
                251314          /),L
                264322          /4,R
                246307          /+,G
                270311          /8,I
                260320          /0,P
                272303          /:,C
                273326          /;,V
                263305          /3,E
                242332          /$,Z
                244304          / ⊃D
                277302          /?,B
                211323          /BELL,S
                266331          /6,Y
                241306          /↑,F
                257330          //,X
                255301          /-,A
                262327          /2,W
                247312          /  ,J
                377377          /FIGURES
                267325          /7,U
                261321          /1,Q
                250313          /(,K
                377377          /LETTERS

START

OCTAL PRINT SUBROUTINE 9-26-62

/ENTER HERE TO SUPPRESS INITIAL ZEROS

OPT,            0
                DAC O→CN
                LAC OP3 1
                JMS OP1
                JMP I OPT
```

```
/ENTER HERE TO SUPPRESS INITIAL ZEROS WITH SPACES
OPS,    0
        DAC OCN
        LAC (JMP OP3
        JMS OP1
        JMP I OPS
/PRINT SUBROUTINE
OP1,    0
        DAC OPM 3
        LAM -5
        DAC O→PC
        LAC (SZA
        DAC OPM
OP0,    LAC OCN
        RCL
        RTL
        DAC OCN
        RAL
        AND (7
OPM,    XX
        JMP OP2
        ISZ OPC
        XX
        TDIGIT
        JMP I OP1
OP2,    TDIGIT
        LAC OPM 1
        ISZ OPC
        JMP OP0-1
        JMP I OP1
OP3,    TSP
        JMP OP0
/END OCTAL PRINT SUBROUTINE

START
```

A	145	DELAY	445
APTEM	67	DIPT	601
AP2	70	DLOUT	454
AP3	71	DRSA	600
B	403	EM0	602
BIP	72	EM02	612
BIRD	204	EM04	625
BTATAB	1177	EM05	636
COMC	247	EM30	653
DCAF	577	EM31	663

EM32	675
EM33	720
ERP	263
ER0	51
ER02	45
ER05	30
ER1	252
ER20	421
ER30	334
ER31	327
ER32	232
ER33	23
E0P	56
E01	153
E02	47
E03	226
E04	246
E05	32
E1H	260
E10	166
E11	174
E12	211
E13	217
E20	443
E20H	442
E20P	431
E30	337
E31	332
E32	236
E33	26
FEED	341
IAWAY	33
INRA	154
INREB	201
IRET	240
NCT	1143
NESQ	407
OCL	1156
OCN	1302
OCS	1157
OCU	1155
OPC	1301
OPM	1264
OPS	1243
OPT	1236
OP0	1256
OP1	1250
OP2	1272
OP3	1277
ORB	700102
OTY	62
PALF	106
PEND	126
PICT	273
PIX	363
PUBS	117
PULS	131
PUNT1	100
PUNT2	351
PUSE	357
P1T	455
P2T	472
RACS	274
RAC2	306
RALPH	161
RECH	412
REFT	36
REO	54
RET1	140
RET2	375
REX	424
RIP	507
RL6	1136
R1T	464
R2T	501
STEM	272
TAPH	315
TBC	1304
TCR	101026
TCRA	1173
TDIGIT	101107
TEMY	1305
TEMY1	1303
TESC	223
TIN	101073
TOUT	255
TSP	101042
TSR	101120
TTAB	10
TYEXIT	601132
TYSVAC	1306
TYT	101054
TY1	100762
TY1A	770
TY1B	1160
TY1C	1167
TY1D	1165
TY2	1012
TY3	101014
XIT	100000

OPM	1264	TY1	100762
OP2	1272	TY3	101014
OP3	1277	TCR	101026
OPC	1301	TSP	101042
OCN	1302	TYT	101054
TEMY1	1303	TIN	101073
TBC	1304	TDIGIT	101107
TEMY	1305	TSR	101120
TYSVAC	1306	TYEXIT	601132
XIT	100000	ORB	700102

11. DIAGRAMS (Not Applicable)

12. REFERENCES (Not Applicable)